

גירסה 1.00 - 8.3.2002



סקריפטים ל-mIRC

ניר אדר

מסמך זה הורד מהאתר www.underwar.co.il

אין להפיץ מסמך זה במדיה כלשהי, ללא אישור מפורש מאת המחבר. מחבר המסמך איננו אחראי לכל נזק, ישיר או עקיף, שיגרם עקב השימוש במידע המופיע במסמך, וכן לנכונות התוכן של הנושאים המופיעים במסמך. עם זאת, המחבר עשה את מירב המאמצים כדי לספק את המידע המדויק והמלא ביותר. כל הזכויות שמורות לניר אדר

במסמך זה נציג אחד מהנושאים הקשורים לכתיבת mIRC Scripts. mIRC זהו לקוח ה-IRC הפופולרי ביותר. בין הסיבות לפופולריות שלו: תאימותו למערכת ההפעלה Windows, נוחות השימוש בו (כפתורים, דיאלוגים, סרגלי כלים) והאפשרות להתאים אותו לצרכי המשתמש. mIRC Scripts באו על מנת לאפשר לאדם לקבל פונקציונליות אשר לא קיימת במקורה ב-mIRC, וגם על מנת לאפשר למשתמש לנצל טוב יותר פונקציות קיימות ב-mIRC. למשל: כתיבת טקסט בצבעים שונים במהירות, העפה מסיבית של אנשים מחדר (במקרה שמתמש הסקריפט הוא "אופ") וכו'. מסמך זה איננו בא ללמד mIRC Scripts מההתחלה. ההנחה היא שהקורא כבר יודע לכתוב סקריפטים ברמה טובה, ורוצה ללמוד כעת את נושא הדיאלוגים ב-mIRC. במסמך זה לא מוסבר על סגנון הכתיבה של mIRC Scripts (הזחה, הערות), על פקודות בסיסיות, על עבודה עם משתנים וגם לא על הגבה לאירועים של המשתמש (פרט לאלו הספציפיים לדיאלוגים), למרות שנתייחס לנושאים אלו.

דיאלוגים

mIRC מאפשר לך ליצור דיאלוגים. דיאלוג זה חלון המשלב בתוכו פקדים (כגון תיבות טקסט, כפתורים, כפתורי בחירה וכו') שבו המשתמש יכול להזין מידע ולקבל מידע.

mIRC אינו מכיל עורך דיאלוגים ויזואלי, והוספת הפקדים ומיקומם נעשית באמצעות כתיבת קוד. עיצוב הפקדים בדיאלוג נעשה באזור ה-remote. מגדירים "טבלה", שהיא למעשה הגדרה של איך יראה הדיאלוג. בעזרת הפקודה dialog אתה מגדיר טבלה ששמה name, בצורה הבאה:

```
dialog name {
  title "your_title_here"
  size x y w h
}
```

הגדרנו "טבלה" בשם name. לאחר מכן קבענו את הכותרת של הדיאלוג החדש ל- your_title_here באמצעות ההצהרה:

```
title "your_title_here"
```

size מגדיר את המיקום והגודל של הדיאלוג שאנו יוצרים. x קובע את ההזחה על ציר x של הדיאלוג. y קובע את ההזחה על ציר y של הדיאלוג. w זהו רוחב הדיאלוג. h זהו גובהו של הדיאלוג. אם x ו y יקבלו ערכים של -1, החלון יהיה במרכז המסך. לאחר שיצרנו חלון, אנחנו רוצים לשים בתוכו פקדים ולעצב אותו.

פקדי קלט ופלט

הקבוצה הראשונה של פקדים היא פקדים שמגדירים אותם בצורה הבאה:

```
<Control Type> "text", id, x y w h, style
```

Control Type הוא סוג הפקד שאנו רוצים ליצור. אפשרויות הבחירה שלנו הם כפתורים (button), תוויות טקסט (text), תיבות טקסט (edit), כפתורי רדיו (radio) וכפתורי בחירה (check). text הוא הטקסט שאנחנו רוצים שיופיע על הפקד. id הוא מספר ייחודי לכל פקד בדיאלוג. במספר זה תשתמש על מנת לבצע פעולות על הפקד בהמשך. x y w h קובעים את מיקומו וגודלו של הפקד בדיאלוג. הפרמטרים הם ארבעה ערכים מספריים ללא פסיקים ביניהם. style קובע את הסגנון של הפקד. נתייחס בהמשך לסגנונות השונים תוך כדי תיאור ספציפי של הפקדים.

פקדי רשימה

הקבוצה השניה של פקדים היא פקדי רשימה, פקדים המכילים רשימות של נתונים. מגדירים אותם כך:

```
<Control Type> id, x y w h, style
```

הפקדים מקבוצה זו הם רשימה (list) ותיבת בחירה (combo), ובהן אין טקסט התחלתי.

פקדים נוספים

1. תמונה

הפקד icon המשמש להצגת קצבי תמונה, המוגדר כך:

```
icon id, x y w h, filename, style
```

filename הוא שם הקובץ שאותו אנו רוצים להציג.

2. קישור

בעזרת link, ניתן להוסיף קישור לדיאלוג. קישור זהו טקסט בצבע כחול בעל קו תחתי, המשנה את צבעו כאשר העכבר עובר מעליו - בדיוק כמו קישורים באתרים. מגדירים קישור בצורה הבאה:

```
link "text", id, x y w h
```

3. קופסא

קופסא (box) היא מסגרת דקה שניתן להוסיף מסביב לפקדים אחרים, על מנת לגרום לדיאלוג שאנו מכינים להראות מאורגן יותר. ניתן גם לשים כותרת שתופיע בראש הקופסא, וכך לתת הגדרה לקבוצת פקדים אחרים שיהיו בתוך הקופסא.

```
box "text", id, x y w h, style
```

text הוא הכותרת שתוצג בראש הקופסא.

4. טאבים

כאשר תיצור דיאלוגים מורכבים, בשלב כלשהו עלול להתעורר הצורך להציג כל פעם רק חלק מהפקדים, מטעמי אסתטיקה ועל מנת שלא להעמיס את המסך בעודף פקדים. לפתרון בעיה זו בא הפקד tab. פקד זה הוא כמו תפריט לחלקים מסוימים בחלון. בעזרת פקד זה ניתן להגדיר אזור שיחולק לעמודות בעלות כותרת. לחיצה על אחת מהכותרות הנ"ל תגרום להצגת הפקדים הקשורים אליה. תוכנות רבות משתמשות בפקד דומה לזה. למשל, קביעת האפשרויות השונות ב Microsoft Word נעשית על ידי חלון, בו ישנם טאבים רבים בעזרתם בוחר המשתמש את האפשרויות אותן הוא רוצה לשנות. צורת ההגדרה של tab היא:

```
tab "text", id, x y w h
tab "text", id
```

פקד ה-tab הראשון שאנו מגדירים קובע את הגודל של הפקד. כל tab שנוסיף אחריו, יצטרף אל פקד זה בתור עמודה נוספת. text קובע את כותרת העמודה. דוגמא מלאה של טאבים נמצאת בהמשך המסמך.

5. סגנונות

לפקדים רבים ניתן להגדיר "סגנון". בנוסף לסגנונות המיוחדים לכל פקד, עליהם נסביר בהמשך המסמך, ישנם סגנונות המשותפים לכל הפקדים, והם: disable – סגנון זה מגדיר את הפקד כלא פעיל. ניתן לראות את הפקד על הדיאלוג, אולם לא ניתן ללחוץ עליו.
hide – סגנון זה מגדיר את הפקד כבלתי נראה. הפקד קיים אולם הוא לא יוצג על הדיאלוג.
group – סגנון זה מגדיר את הפקד כראשון מקבוצה. כל הפקדים המופיעים אחרי הפקד הזה ועד לפקד הבא שסגנונו הוא group מוגדרים כקבוצה אחת.

נדגים את השימוש בפקדים באמצעות מספר דוגמאות.

דוגמא 1 – שימוש בתוויות טקסט, כפתורים וכפתורי רדיו

נשים את הקוד הבא בremote:

```
dialog dlgFirstDialog {
  title "My First Dialog"
  size 164 130 290 270

  button "Close",1,113 209 73 27,cancel
}
```

הגדרנו טבלה בשם dlgFirstDialog. קבענו את הכותרת שלה ל- "My First Dialog" על ידי הפקודה "title" והגדרנו את הגודל של הדיאלוג על ידי הפקודה "size".
לאחר מכן הגדרנו כפתור על ידי הפקודה "button", קבענו שהטקסט שיוצג בו יהיה "Close". קבענו את id שלו ל- 1. הגדרנו את המיקום ואת הגודל שלו בדיאלוג וקבענו את הסגנון שלו בתור cancel.
לכל דיאלוג חייב להיות כפתור אחד שהסגנון שלו יוגדר בתור cancel או בתור ok. כפתור שהסגנון שלו הוא אחד מהנ"ל, יגרום לסגירת הדיאלוג אחרי שלוחצים עליו. כמו כן, לחיצה על המקש esc נחשבת כלחיצה על כפתור שהוגדר כcancel ולחיצה על Enter נחשבת כלחיצה על כפתור ok. סגנון נוסף שיכול להיות לכפתור הוא default. כאשר כפתור מוגדר בסגנון זה, כפתור זה יהיה הכפתור שילחץ במקרה וילחץ המקש Enter, ולא כפתור ה- ok.

/dialog

על מנת שנוכל להפעיל את הדיאלוג שכתבנו עלינו להשתמש בפקודה /dialog פקודה זו מבצעת שני סוגים עיקריים של פעולות. הסוג הראשון הוא הצגת הדיאלוג שיצרנו על המסך, בהתאם לפרמטרים שאנו מספקים לה. הסוג השני של פעולות הוא שינוי תכונות כגון כותרת, מיקום וכדומה של דיאלוגים שכבר מופיעים על המסך.
על מנת להציג דיאלוג על המסך, עלינו להשתמש בתחביר הבא:

```
/dialog -m[dno] name table
```

name זהו השם בו נפנה לדיאלוג בהמשך. table זהו שם הטבלה שהגדרנו המכילה את הפקדים שאנו רוצים לשים בדיאלוג.
הפרמטר m הוא הכרחי, והוא מודיע ל- mIRC שאנו רוצים להציג דיאלוג חדש על המסך. הפרמטרים d, n, o הם אופציונליים. ניתן להשתמש בהן ביצירת החלון וגם כאשר הוא כבר קיים. נסביר על פרמטרים אלו בהמשך.

הפקודה /dialog משמשת גם לבצע פעולות על הדיאלוג לאחר שיצרנו אותו. כדי לבצע פעולות על הדיאלוג לאחר שיצרנו אותו, נשתמש בתחביר הבא:

```
/dialog -[donxkcv] name
```

name הוא השם שהענקנו לחלון כאשר יצרנו אותו. אנחנו מעבירים אותו כפרמטר כדי שה- mIRC ידע לאיזה חלון לגשת ולשנות.

מצב החלון

שימוש בפרמטר d מגדיר שהדיאלוג יפתח בתור חלון נפרד (שיוצג בשורת המשימות של Windows) ולא תחת החלון של mIRC. הפרמטר o מגדיר את הדיאלוג בתור always on top. דיאלוג מסוג זה תמיד יראה מעל כל החלונות, ללא קשר לחלון הפעיל כרגע. הפרמטר n קובע שהדיאלוג לא יהיה מסוג always on top.

בעזרת הפרמטר v- אנו הופכים את הדיאלוג לחלון הפעיל, במידה והוא איננו כזה.

סגירת הדיאלוג

שימוש בפרמטר k גורם ללחיצה על הלחצן שסגנונו הוגדר ok ולסגירת הדיאלוג. הפרמטר c גורם ללחיצה על הלחצן שסגנונו הוגדר cancel ולסגירת הדיאלוג. הפרמטר x גורם לסגירת הדיאלוג, מבלי לגרום לשום אירוע להתרחש.

בפרמטרים k ו- c משתמשים מכיוון שבנוסף לכך שאנו רוצים לסגור את החלון כמו שגורם הפרמטר x, אנחנו גם רוצים לבצע את הקוד שמבצעים הכפתורים cancel\ok.

ישנן שתי פעולות נוספות שניתן לבצע על חלון, אשר להן תחביר שונה מעט:

שינוי מיקום הדיאלוג

```
/dialog -s name x y w h
```

בעזרת הפרמטר s אנו מסוגלים לקבוע את מיקומו וגודלו של דיאלוגים לאחר שהם כבר מוצגים על המסך.

שינוי כותרת הדיאלוג

```
/dialog -t name text
```

בעזרת הפרמטר t אנו קובעים את הטקסט שיוצג בכותרת הדיאלוג. text הוא הטקסט החדש שאנו רוצים לשים.

לאחר שהכרנו את הפקודה /dialog, אנו מסוגלים לחזור לדוגמה שהתחלנו קודם לכן. כתוב את השורה הבאה באחד מחלונות הmIRC כדי לראות את הדיאלוג שיצרנו.

```
/dialog -m dlgFirstDialog dlgFirstDialog
```

בפקודה זו אנו מציגים על המסך את הדיאלוג שיצרנו, קובעים שהשם בו נפנה אליו יהיה dlgFirstDialog וכן מציינים שאנו רוצים להשתמש בפקדים שהגדרנו בטבלה dlgFirstDialog בדיאלוג זה. הערכים של ה-name וה-table יכולים להיות זהים, ואף מקובל לתת להם את אותו השם.

נפתח חלון ובו כפתור אחד. בלחיצה על הכפתור, או בלחיצה על המקש esc, החלון ייסגר. נרחיב עכשיו את הדיאלוג שלנו, ונוסיף לו עוד פקדים:

```
dialog dlgFirstDialog {
  title "My First Dialog"
  size 164 130 290 270


  button "Close",1,113 209 73 27,cancel
  text "Please select one of the options",2,64 22 152
  18,center
  radio "Option Number 1",3,22 68 123 21
  radio "Option Number 2",4,22 96 123 21
  radio "Option Number 3",5,22 124 123 21
}
```

הוספנו תווית טקסט (text). תווית טקסט זהו טקסט שמוצג על הדיאלוג בנקודה שנגדיר. המשתמש לא יכול לשנות את תוכן הטקסט. את תוכן הטקסט ניתן לשנות רק באמצעות קוד. תוכן הטקסט ההתחלתי של התווית הוא "Please select one of the options" והסגנון שהגדרנו לטקסט הוא center, כלומר ממורכז. אם לא מוגדר סגנון לתווית טקסט, תוכן התווית מיושר לשמאל. ניתן להגדיר מרכז בעזרת center ויישור לימין באמצעות הסגנון right. לאחר מכן הגדרנו שלושה כפתורי רדיו וקבענו את מיקומם ואת הטקסט שיופיע עליהם.

שוב, נשתמש בפקודה

```
/dialog -m dlgFirstDialog dlgFirstDialog
```

על מנת להציג את הדיאלוג על המסך.



אירועי Dialog

לאחר שעיבנו את הדיאלוג, אנו רוצים לגרום לו להגיב למשתמש. כאשר משתמש לוחץ על פקד, סוגר את הדיאלוג או מבצע פעולה כלשהי הקשורה לדיאלוג, מתרחש אירוע מסוג Dialog. אנו מגיבים לאירוע זה בצורה הכללית הבאה:

```
on 1:dialog:name:event:id: {
}
```

name הוא שם הדיאלוג שאנו רוצים להגיב לאירועים שלו.
 event הוא האירוע שאנו רוצים להגיב אליו. ערכים אפשריים לפרמטר זה הם:
 - init – אירוע זה נקרא כאשר הדיאלוג נוצר.
 - edit – אירוע זה נקרא כאשר טקסט בתוך תיבת טקסט משתנה על ידי המשתמש.
 - sclick – אירוע זה נקרא כאשר המשתמש לוחץ לחיצה בודדת עם העכבר על פקד כלשהו.
 - dclick – אירוע זה נקרא כאשר המשתמש לוחץ לחיצה כפולה עם העכבר על מקש כלשהו.

בתוך אירוע זה ניתן להשתמש בפונקציות הבאות, על מנת לקבל מידע על האירוע:
 \$dname מחזירה את שם הדיאלוג שבו התרחש האירוע.
 \$devent מחזירה את שם האירוע שהתרחש.
 \$did מחזירה את id של האירוע שהתרחש.
 פרמטרים אלו רלוונטיים, כאשר במקום להתייחס לדיאלוג, אירוע או id ספציפי אנו משתמשים ב*.

נחזור עכשיו לדיאלוג שלנו. נגרום להופעת הודעה ב-Status Window, במקרה שבחרים אחד מכפתורי הרדיו:

```
on 1:dialog:dlgFirstDialog:sclick:*: {
  if ( $did == 3 ) echo -s Option 1 Selected
  if ( $did == 4 ) echo -s Option 2 Selected
  if ( $did == 5 ) echo -s Option 3 Selected
}
```

האירוע שאנו מגיבים אליו הוא sclick. בעזרת * אנו קובעים שהאירוע יתפוס את כל הלחיצות בדיאלוג, ולא מגדירים id מסוים שייתפס.
 אנו בודקים בעזרת הפונקציה \$did מהו id של הפקד הנלחץ ומדפיסים הודעה ב-Status Window על האפשרות שנבחרה.

דוגמא 2 – טאבים

כפי שהסברנו מקודם, ניתן להגדיר אזור שיחולק לעמודות בעלות כותרת על ידי הפקודה tab. לחיצה על אחת מהכותרות הנ"ל תגרום להצגת הפקדים הקשורים אליה. לאחר שיצרנו את העמודות, אנחנו רוצים לשייך פקדים שונים לעמודות אלו. על מנת לעשות זאת, מוסיפים בסוף ההגדרה של הפקד המבוקש את המילה tab, ולאחריה את id של העמודה אליה אנו רוצים שהפקד ישתייך.

הדוגמא הבאה יוצרת דיאלוג בו אזור שחולק ל 4 עמודות. כותרת העמודה הראשונה היא M, השניה היא I, השלישית היא R והרביעית היא C. בכל עמודה מוצג כפתור אחד עליו נמצא טקסט. נשים את הקוד הבא ב-remote:

```
dialog dlgMIRC {
  title "mIRC"
  size -1 -1 210 145

  tab "m", 1, 5 5 200 130
  tab "I", 2
  tab "R", 3
  tab "C", 4

  button "m is for ... ;)", 11, 55 50 100 48, ok tab 1
  button "I is for Internet", 12, 55 50 100 48, tab 2
  button "R is for Relay", 13, 55 50 100 48, tab 3
  button "C is for Chat", 14, 55 50 100 48, tab 4
}
```

נציג את הדיאלוג באמצעות הפקודה

```
/dialog -m dlgMIRC dlgMIRC
```

ונראה איך כל פעם מוצג פקד אחר, בהתאם לעמודה עליה אנו לוחצים.



שינוי הדיאלוג לאחר שנוצר

לאחר שיצרנו את הדיאלוג, אנו נרצה לשנות לעיתים את הדברים המופיעים בו. סיבות אפשריות לשינויים הם הגבה לאירועים בIRC, הגבה ללחיצות מקש/עכבר של המשתמש ועוד. אנו מסוגלים לבצע פעולות שונות בדיאלוג בעזרת הפקודה /did. המבנה הכללי של הפקודה הוא

```
/did -ftebvhnmcukradiogj name id [n] [text | filename]
```

name הוא שם הדיאלוג. id זהו id של הפקד עליו אנו רוצים לבצע את הפעולה. הפרמטרים האחרים אופציונליים. בעזרת הפרמטר f אנו מסוגלים לקבוע פוקוס על אחד מהפקדים, כלומר לסמן אותו בתור הפקד הפעיל כרגע.

לדוגמא:

אם היינו כותבים בדוגמא הקודמת

/did -f dlgMIRC 12

היינו הופכים את הכפתור בעל הכותרת "I is for Internet" לכפתור הפעיל.

הפרמטר t קובע את הפקד הנבחר בתור ה-default. (הפקד שהוגדר בזמן יצירת הדיאלוג כ-default יפסיק להיות default אם מוגדר פקד אחר).

הפרמטר e משמש כדי להפוך פקד לפעיל, במידה ולא היה כזה. מולו הפרמטר b משמש כדי להפוך פקד ללא פעיל, כלומר, שהמשתמש לא יוכל ללחוץ עליו. אם היינו כותבים בדוגמא הקודמת

/did -b dlgMIRC 12

הכפתור היה הופך ללא פעיל. הטקסט עליו היה נהפך לאפור, והמשתמש לא היה יכול ללחוץ עליו.

הפרמטר h משמש להפיכת פקד לבלתי נראה. הפרמטר v יהפוך פקדים בלתי נראים לנראים. לתיבות טקסט (edit) יש אפשרות נוספת – להפוך את הטקסט שלהן לניתן לשינוי או לא ניתן לשינוי על ידי המשתמש. בעזרת הפרמטר m הופכים את הטקסט ללא ניתן לשינוי ובעזרת הפרמטר n מחזירים את האפשרות לשנותו. ההבדל בין אפשרות זו לאפשרות לעשותן לא פעילות, היא שכאשר התיבה מבוטלת באמצעות הפרמטר m המשתמש אינו יכול לשנות את הטקסט, אולם הוא יכול להעתיק (copy) את הטקסט ולסמן אותו, ואילו כאשר התיבה מבוטלת עם הפרמטר b לא ניתן לגעת בה כלל.

הפרמטר c משמש לבחירה בכפתור רדיו (radio), לסימון כפתור בחירה (check), ולבחירת שורה בתוך תיבת בחירה (combo) או ברשימה (list).

לדוגמא:

ההנחה ש id מספר 10 הוא id של כפתור בחירה בדיאלוג שלנו, הפקודה

/did -c Dialog_Name 10

תסמן את הכפתור בחירה.

אם id מספר 11 הוא תיבת בחירה, הפקודה הבאה תבחר את האיבר השלישי ברשימה:

/did -c Dialog_Name 11 3

הפרמטר u משמש לביטול הבחירה של כפתור רדיו, כפתור בחירה, תיבת בחירה או רשימה.

הפרמטר z משמש למחיקת כל הטקסט הנמצא על פקד נבחר. הפרמטר a משמש להכנסת טקסט בסוף הטקסט הקיים על הפקד (בחלק מהפקדים כגון כפתורים, כפתורי רדיו, תוויות טקסט - פרמטר זה גורם להחלפת הטקסט, ולא להוספה בסופו). הפרמטר d בשילוב מספר, גורם למחיקת השורה בעלת אותו מספר מהפקד. למשל, כדי למחוק את השורה השלישית מתוך תיבת בחירה בעלת id של 15, נכתוב

/did -d Dialog_Name 15 3


הפרמטר i בצירוף מספר משמש להכנסת שורה חדשה לפקד בשורה המבוקשת. הפרמטר o בצירוף מספר משמש להחלפת השורה המבוקשת בפקד בטקסט חדש.

דוגמא 3 – תיבות רשימה

בדוגמא זו ניצור דיאלוג עם 2 כפתורים – Add, ו-Remove. כמו כן ניצור תיבת טקסט אחת, רשימה אחת וכפתור שימש לסגירת הדיאלוג. כאשר המשתמש ילחץ על add נוסף את הטקסט שבתחת הטקסט לרשימה. כאשר המשתמש ילחץ על Remove, נבדוק האם נבחר פריט ברשימה, ואם כן נמחק אותו. נלמד תוך כדי הדוגמא על הפונקציה \$did והפרמטרים השונים שלה, המאפשרים לנו לקבל מידע על הפקדים השונים שיצרנו.

דבר ראשון, נעצב את הדיאלוג:

```
dialog dlgLists {
  title "Lists Work"
  size 205 126 250 240
  list 1,143 24 87 154
  button "Add",2,39 42 65 22
  button "Remove",3,39 71 65 22
  edit "Text",4,31 111 81 20
  button "Close",5,87 191 72 29,cancel, default
  box "" ,6,16 17 110 154
}
```



לצורך ביצוע הדוגמא, עלינו להיות מסוגלים לקבל מידע על הפקדים שיצרנו. למשל, עלינו לדעת איזה, אם בכלל, פריט ברשימה נבחר כרגע. כמו כן נרצה לדעת מה הטקסט הנוכחי בתיבת העריכה, על מנת שנוכל להוסיף אותו לרשימה. לעזרתנו באה הפונקציה \$did.

צורת הפנייה לפונקציה הזו היא \$did(name,id,N) . name הוא שם הדיאלוג בו נמצא הפקד. id זהו id של הפקד. N הוא פרמטר אופציונלי, ואנו מוסיפים אותו אם אנו רוצים להתייחס לשורה N ברשימה או בתיבת בחירה. על מנת לגשת לתכונות השונות של כל פקד, מוסיפים . (נקודה) אחרי הפונקציה \$did ואז את הערך המבוקש לנו, מתוך הערכים הבאים:

text – מחזיר את הטקסט המופיע על הפקד בשורה הנוכחית או בשורה N אם מוגדר N. למשל,

```
$did(dlgLists, 2).text
```

יחזיר לנו "Add".

אם לא קיים טקסט, נקבל בחזרה את הערך null.

len – מחזיר לנו את אורך הטקסט המופיע בשורה הנוכחית או בשורה N

```
$did(dlgLists, 2).len
```

יחזיר לנו 3. (בטקסט "Add" יש 3 אותיות).

lines – פרמטר זה יחזיר לנו את מספר שורות הטקסט שיש בפקד.
 sel – מחזירה את מספר השורה שנבחרה ברשימה או בתיבת בחירה.
 seltext – מחזיר את הטקסט הנוכחי שנבחר בפקד כגון רשימה או תיבת בחירה.
 state – מחזיר את מצב הפקד (בפקדים שיש בהם מצבים, שהם כפתורי רדיו וכפתורי בחירה). המצב יכול להיות 0 (off), 1 (on) או 2 (לא מוגדר).

visible – מחזיר ערך \$true אם הפקד נראה, או \$false אם הוא בלתי נראה.
 enabled – מחזיר ערך \$true אם הפקד פעיל, או \$false אם הוא לא פעיל.

נחזור כעת לדוגמא שלנו.

אם נביט, נראה כי בדיאלוג שיצרנו, מוצג הטקסט Text בתיבת עריכה בהתחלה. אנו רוצים שכאשר הדיאלוג יפתח, לא יהיה טקסט בתיבה, והמשתמש יוכל להקיש את הטקסט שהוא רוצה. נגיב לאירוע של הדיאלוג, ונמחוק את הטקסט מהתיבה:

```
on 1:dialog:dlgLists:init:*: {
  did -r dlgLists 4
}
```

עכשיו נטפל בלחיצה על המקש Add.

כאשר המשתמש ילחץ על המקש, הסקריפט יבדוק האם קיים טקסט בתוך התיבה, ואם כן, יוסיף את הטקסט לרשימה, וינקה את תיבת הטקסט. כדי להגיב ללחיצה על Add אנו צריכים להגיב לאירוע sclick.

```
on 1:dialog:dlgLists:sclick:*: {
  if ($did == 2 ) {
    if ( $did(dlgLists, 4).text != $null ) {
      did -a dlgLists 1 $did(dlgLists, 4).text
      did -r dlgLists 4
    }
  }
}
```

כאשר המשתמש לוחץ במקום כלשהו בדיאלוג, הסקריפט בודק האם id שנלחץ הוא 2. אם כן אנו בודקים בעזרת (\$did(dlgLists, 4).text != \$null) האם ישנו טקסט בתיבת הטקסט. אם כן, אנו מוסיפים את הטקסט אל הרשימה, ואז מוחקים את הטקסט.

לאחר שגרמנו לכפתור Add לעבוד, אנו רוצים לגרום גם לכפתור Remove לעבוד. כאשר הכפתור יילחץ, הסקריפט יבדוק האם המשתמש בחר את אחד מפריטי הרשימה. אם כן, פריט זה יימחק מהרשימה. הנה הקוד החדש של האירוע sclick:

```
on 1:dialog:dlgLists:sclick:*: {
  if ($did == 2 ) {
    if ( $did(dlgLists, 4).text != $null ) {
      did -a dlgLists 1 $did(dlgLists, 4).text
      did -r dlgLists 4
    }
  }
  if ($did == 3 ) {
    if ( $did(dlgLists, 1).sel != $null ) {
      did -d dlgLists 1 $did(dlgLists, 1).sel
    }
  }
}
```

אנו בודקים האם ישנו פריט מסומן : (\$did(dlgLists, 1).sel != \$null) במידה וכן, אנו מוחקים מהרשימה את הפריט שהמספר הסידורי שלו הוא \$did(dlgLists, 1).sel (הפריט הנבחר).

סיימנו את כתיבת הקוד של הדוגמא. נציג את הדיאלוג על המסך בעזרת הפקודה.

```
/dialog -m dlgLists dlgLists
```

נוסיף איברים לרשימה, נמחק אותם ונראה איך הקוד עובד.